SOFTWAREŮPRACTICE AND EXPERIENCE, VOL. 0(0), 1Ű30 (MONTH 1998)

Unparsing Expressions With Pre�x and Post�x Operators

Norman Ramsey

Department of Computer Science, University of Virginia, Charlottesville, VA 22903

August 13, 1998

SUMMARY

Unparsing is the problem of transforming an internal representation of a program into an external, concrete syntax. In conjunction with prettyprinting, it is useful for generating readable programs from internal representations. If the target language uses pre�x and post�x operators, the problem is nontrivial. This paper shows how to unparse expressions using a simple, bottom-up tree walk, which keeps track of the least tightly binding operator not enclosed by parentheses. During the tree walk, this operator is compared with the operator of the parent expression, and parentheses are inserted based on the precedence, associativity, and �xity (in�x, pre�x, or post�x) of the two operators. The paper is a literate program. It includes code for the unparser and for its inverse parser, both of which can handle operators of dynamically chosen precedence and associativity. Supporting such operators is useful for languages like ML, in which programmers may assign precedence and associativity to their own functions.

KEY WORDS

KEY WORDS Unparsing Parsing Prettyprinting Literate Programming Standard ML

1. INTRODUCTION

Automatic generation of computer programs from high-level speci^cations is a well-known technique, which has been used to create scanners, parsers, code generators, tree rewriters, packet ^lters, dialogue handlers, machine-code recognizers, and other tools. To make it easier to debug an application generator, and to help convince prospective users that the generated code can be trusted, the generated code should be as idiomatic and readable as possible. It is therefore desirable, and sometimes essential, that the generated code use in^x, pre^x, and post^x operators, and that expressions be presented without unnecessary parentheses. Another helpful technique is to use a prettyprinter to automate indentation and line breaking.^{1, 2} This paper presents a method of automatically parenthesizing expressions that use pre^x, post^x, and implicit operators; the method is compatible with automatic prettyprinting. The paper also shows how to parse expressions that use such operators. The parsing algorithm can be used even if operator precedences are not known at compile time, so it can be used with arbitrarily many user-de^ned precedences.

This paper is a Şliterate program,Ť^3 which means that a single source ^le is used to prepare both manuscript and code. Including code provides a formal, precise, testable statement of the unparsing

CCC 0038Ű0644/98/000001Ű30 Received 29 August 1997

cfl1998 by John Wiley & Sons, Ltd. Revised 27 March 1998

2 NORMAN RAMSEY

algorithm. The code is written in the purely functional subset of Standard ML.^4 ML is a good choice for several reasons. Referential transparency makes the subset easy to reason about, so we can show the correctness of simple unparsing algorithms. It is possible to express some properties of the data as type properties, which are checked by the compiler. Finally, SML’s parameterized modules make the code reusable in several different contextsŮthe code that appears in the paper is the code used in both the SLED and the λ-RTL compilers,^{5, 6} for example. The paper explains many ML constructs as they are used, but readers not familiar with ML may still wish to consult References 7Ű9 for an introduction.

The paper presents algorithms for two versions of the unparsing problem. In the ^rst version, all operators are binary in^x operators; this version of the unparsing problem is well understood, and it is simple enough to make a proof of correctness tractable. The invariants and insights from this proof lead to an algorithm for a more general version, which handles pre^x and post^x operators, as well as nonassociative in^x operators of arbitrary arity. The algorithm can also handle some Şmix^xŤ operators, like the array-indexing operator ·[·]. The algorithm is the main contribution of this paper. It is presented as executable code, which uses parameterized modules to make it independent of the representation of the unparsed form.

2. PARSING AND UNPARSING

Parsers and unparsers translate between concrete and abstract syntax. The concrete syntax used to represent expressions is usually ambiguous; ambiguities are resolved using operator precedence and parentheses. Internal representations are unambiguous but not suitable as concrete syntax. For example, expressions are often represented as simple trees, in which an expression is either an ŞatomŤ (constant or variable) or an operator applied to a list of expressions. We can describe such trees by the following grammar, written in extended BNF,^{10} in which nonterminals appear in italic and terminals in slant fonts:

expression ⇒ atom | operator { expression } LISP and related languages use a nearly identical concrete syntax, simply by requiring parentheses around each application. The designers of other languages have preferred other concrete syntax, using in^x binary operators, pre^x and post^x unary operators, and possibly Şmix^xŤ operators, as described by the following grammar:

expression ⇒ atom | expression in�x-operator expression | pre�x-operator expression | expression post�x-operator | expression { nary-operator expression } | other form of expression. . .

Even if the internal form distinguishes these kinds of operators, this grammar is unsuitable for parsing, because it is ambiguous. Grammars intended for parsing use various levels of precedence and associativity to resolve ambiguities. Once precedence and associativity have been determined, they can be encoded by introducing new nonterminals for each level of precedence. Users of shift-reduce parsers can specify precedence and associativity directly, then have the parser use them to resolve shift-reduce con~icts.

The parsing problem is to translate from concrete syntax to internal representation. The unparsing problem is to take an internal representation and to produce a concrete syntax that, when parsed, reproduces the original internal representation. That is, an unparser is correct only if it is a right inverse of its parser. For example, if our internal representation is a tree representing the product of x + y with z, we cannot simply walk the tree and emit the concrete representation x + y × z; we must emit (x + y) × z instead.

As stated, the unparsing problem does not have a unique solution; in general, there are many possible concrete syntaxes that, when parsed, reproduce the original internal representation. For example, we could choose a concrete syntax that parenthesizes every expression, as in (((x) + (y)) × (z)). The LISP rule parenthesizes every application, as in ((x + y) × z). Rules like these use too many parentheses. Code with extra parentheses is unidiomatic and unreadable, like this C code, which has been emitted using something like the LISP rule:

if ((((Mem.u).Index8).index) != 4)
  if (!((unsigned(((Mem.u).Index8).d)) < 0x100))
    fail((("((Mem.u).Index8).d = 0x%x won’t fit in 8 unsigned bits"),
           (((Mem.u).Index8).d)));
  else
    {
      emit((((7 | (15 << 4)) | (1 << 3)), 1));
      emit((((4 | (3 << 3)) | (1 << 6)), 1));
      emit((((((((Mem.u).Index8).ss) << 6) |
             ((((Mem.u).Index8).index) << 3)) |
             (((Mem.u).Index8).base)), 1));
      emit((((((Mem.u).Index8).d) & 0xff), 1));

    }
else
  fail(("Conditions not satisfied for constructor CALL.Epod"));

The unparser in this paper is not merely a parser’s right inverse; it also produces results without extra parentheses:

if (Mem.u.Index8.index != 4)
  if (!((unsigned)Mem.u.Index8.d < 0x100))
    fail("Mem.u.Index8.d = 0x%x won’t fit in 8 unsigned bits",
         Mem.u.Index8.d);
  else
    {
      emit(7 | 15 << 4 | 1 << 3, 1);
      emit(4 | 3 << 3 | 1 << 6, 1);
      emit(Mem.u.Index8.ss << 6 | Mem.u.Index8.index << 3 |
           Mem.u.Index8.base, 1);
      emit(Mem.u.Index8.d & 0xff, 1);
    }
else
  fail("Conditions not satisfied for constructor CALL.Epod");

The unparsing method uses information about precedence, associativity, and Ş^xityŤ of operators to transform an internal form into a concrete syntax. The method works from the bottom up and from the inside out. In each expression, it ^nds the operator, and it considers the subexpressions and their positions, left or right, relative to the operator. It decides whether to parenthesize subexpressions by comparing precedence and associativity of the current operator with the precedences and associativities

4 NORMAN RAMSEY

of the most loosely binding operators in the subexpressions. Operators that are ŞcoveredŤ by parentheses do not participate in the comparison. Here are four example trees: × + x y z + × x y + z w post^x ++

*
p pre^x ++

*
p

(a) (b) (c) (d) In tree (a), discussed above, the unparser renders the left subtree as x + y, without parentheses, because the subexpressions x and y contain no operators. At the root, because operator × has higher precedence than operator +, the unparser parenthesizes the left subexpression, and the ^nal result is (x + y) × z. In tree (b), + has lower precedence than ×, so it is not necessary to parenthesize the left subtree. Because + is left-associative, however, it is necessary to parenthesize the right subtree, because it is a right child of a parent with the same precedence and left associativity. The ^nal result is therefore x × y + (z + w). Trees (c) and (d) show pre^x and post^x operators as used in C. In tree (c), ++ and * have opposite Ş^xityŤ (one is post^x and one is pre^x), and because ++ has higher precedence than *, it is necessary to parenthesize the subexpression, resulting in (*p)++. In tree (d), ++ and * have the same ^xity, so precedence is irrelevant, and the expression is unparsed as ++*p.

Although the full unparsing method handles pre^x, post^x, and in^x operators, it is easier to understand if we begin with the simpler case in which all operators are binary in^x operators.

3. PARSING AND UNPARSING WITH INFIX OPERATORS

Abstract and concrete syntax for in�x operators

Type rator represents a binary in^x operator, which has a text representation, a precedence, and an associativity: ⟨in�x⟩≡

type precedence = int
datatype associativity = LEFT | RIGHT | NONASSOC
type rator = string * precedence * associativity

This ML code uses simple integers (int) to represent precedence, an enumeration to represent associativity, and a triple to represent an operator. (In the context of an ML type de^nition, a * does not represent multiplication; it connects elements of a tuple.) The more general unparser, presented below, shows how to use an arbitrary type, not just string, as an operator’s concrete representation.

Precedence and associativity determine how in^x expressions are parsed into trees, or equivalently, how they are parenthesized. For example, if operator ⊗ has higher precedence than operator ⊕, then x ⊕ y ⊗ z = x ⊕ (y ⊗ z) and x ⊗ y ⊕ z = (x ⊗ y) ⊕ z. When two operators have the same precedence, associativity is used to disambiguate. If ⊕ is left-associative, x ⊕ y ⊕ z = (x ⊕ y) ⊕ z; if it is right-associative, x ⊕ y ⊕ z = x ⊕ (y ⊕ z). Some languages have non-associative operators; if ⊕ is non-associative, then

(x ⊕ y) ⊕ z ≠ x ⊕ y ⊕ z ≠ x ⊕ (y ⊕ z);

and x ⊕ y ⊕ z may be illegal. The comma that separates parameters in a C function call is an example of a non-associative operator.

Many expressions are obtained by applying operators to other expressions, but there must always be indivisible constituents of expressions. We call such constituents atoms. They appear at the leaves of abstract syntax trees and as the non-operator tokens in concrete syntax. In most languages the atoms include identi^ers, integer and real literals, string literals, and so on. In our initial, simple model, an expression is either an atom or an in^x operator applied to two expressions: ⟨in�x⟩+≡

datatype ast = ATOM of string
             | APP of ast * rator * ast

Type ast represents an expression’s abstract syntax tree. This ML datatype de^nition is something like a production in a context-free grammar; it gives two alternative ways of constructing an ast. These alternatives are always given names (ATOM and APP), which are called constructors. An ML datatype may contain any number of constructors, and there may be data associated with each one. Here, for an atom, we care only about the atom’s string representation. For an application, we want the operator and the asts representing the two operands. Because ast is used in its own de^nition, it is a recursive type, and values of type ast are trees.

Type type ast represents the input to the unparser; we also need a type to represent the unparser’s output. For simplicity, we treat this output as a sequence of lexemes, where a lexeme represents an atom, an operator, or a parenthesis. Moreover, we undertake to emit only sequences in which atoms and operators alternate, or in which parenthesized sequences take the place of atoms. Let us call such a sequence an image and use a representation that forces it to satisfy the following grammar:

image ⇒ lexical-atom { operator lexical-atom } lexical-atom ⇒ atom | | ( image )

In the corresponding ML, the sequence in braces becomes the type image’: ⟨in�x⟩+≡

  datatype image = IMAGE of lexical_atom * image’
  and image’ = EOI (* end of image *)
                   | INFIX of rator * lexical_atom * image’
  and lexical_atom = LEX_ATOM of string
                   | PARENS of image
EOI represents the end of the image, and PARENS represents an image in parentheses. This representa-

tion enforces the invariants that expressions and operators alternate, and that the ^rst and last elements of an image are expressions.

Parsing in�x expressions

To be correct, an unparser must produce a sequence that parses back into the original abstract syntax tree. We develop an unparsing algorithm by thinking about parsing. To understand how to minimize parentheses, we need to consider where parentheses are needed to get the correct parse. Suppose we have an abstract syntax tree that has been obtained by parsing an image without any parentheses. Then wherever the syntax tree has an APP node whose parent is also an APP node, there are two cases: the child may be the left child or the right child of its parent:

   outer

inner

fi fi fi fi

outer

   inner

L L L L

6 NORMAN RAMSEY

Because this tree was obtained by parsing parenthesis-free syntax, either the inner operator has higher precedence than the outer, or they have the same precedence and associativity and the associativity is to the left (right) if the inner is a left (right) child of the outer. We formalize this condition using the predicate noparens(inner, outer, side), where side is LEFT or RIGHT and noparens is de^ned as follows:

⟨in�x⟩+≡ noparens : rator * rator * associativity -¿ bool

fun noparens (inner as (_, pi, ai), outer as (_, po, ao), side) =
           pi > po
    orelse pi = po andalso ai = ao andalso ao = side

where pi, ai, po, and ao stand for precedence or associativity of inner or outer operators.

This ML code de^nes a function noparens. Its name and type are given in the box at the beginning of the chunk. Type assertions in boxes are inserted into the code and checked by the compiler. The type of noparens shows that it is a Boolean function of three arguments (two operators and an associativity). The structure of the de^nition is fun name binding = body. The binding associates names to the arguments; these names are used in the body. inner as ( , pi, ai) is a nested binding; it binds pi to the precedence of inner and ai to the associativity of inner. The underscore is a Şwildcard,Ť which indicates that the text representation of inner is to be ignored and not bound to any name. The process of using the binding to associate names with the arguments and parts of the arguments is called ML pattern matching.

Readers familiar with the treatment of operator-precedence parsing in Section 4.6 of Reference 11 may recognize that

 noparens(i; o; LEFT) ≡ i ·> o
noparens(i; o; RIGHT) ≡ o <· i

We use noparens during unparsing to decide when parentheses are needed and during parsing to decide whether to shift or reduce.

To prove correctness of the unparser, I introduce a simple LR parser. The unparser is correct if

parse (unparse e) = e for any syntax tree e. In other words, if we begin with an abstract

syntax tree e, unparse it, and then parse the resulting lexemes, we get back the original abstract syntax tree.

In the parser’s stack, expressions and operators alternate. Unless the stack is empty, there is an operator on the top, which is written to the right. ⟨in�x⟩+≡

datatype stack = BOT
               | PUSH of stack * ast * rator

This recursive datatype is isomorphic to a list of (operator, ast) pairs, but it is helpful to de^ne a special type because the constructors BOT and PUSH distinguish parser stacks from other kinds of lists.

The parser itself can be simpli^ed by a bit of trickery. Instead of treating Şbottom of stackŤ or Şend of inputŤ as special cases, treat them as occurrences of a sentinel operator, minrator. minrator has precedence minprec, which must be lower than the precedence of any real operator. Using this trick, we de^ne functions that return the operator on the top of the stack and the operator about to be scanned in the input.

⟨in�x⟩+≡ srator : stack -¿ rator

                                                 irator : image’ -¿ rator
val minrator : rator =
 ("<minimum-precedence sentinel>", minprec, NONASSOC)
fun srator (PUSH (_, _, $)) = $
  | srator BOT = minrator
fun irator (INFIX ($, _, _)) = $
  | irator EOI = minrator

Given a stack stack and an input sequence of tokens ipts, we may abbreviate srator stack as ⊕_{stack} and abbreviate irator ipts as ⊕_{ipts}.

These function de^nitions show a new ML constructŮwe de^ne functions on datatypes by pattern-matching against the datatype constructors. We provide a binding and a body for each constructor. This ability to use pattern matching to do case analysis is one of the advantages of using ML, especially because the compiler ensures that all cases are covered.

This code also illustrates an unusual syntactic feature of ML: we can use strings of symbols, as well as strings of letters, to make identi^ers. In this paper, by convention, the identi^er $ stands for an arbitrary operator.

We now have enough machinery to write a shift-reduce LR parser, which continues parsing until its stack and input are empty. The parser’s state includes a stack with an operator ⊕_{stack} on top, a current expression e, and the input, which begins with the operator ⊕_{ipts}. This state is represented by three parameters to the function parse’. Parsing is complete when the stack is empty (BOT) and the input is empty (EOI).

⟨in�x⟩+≡ parse’ : stack * ast * image’ -¿ ast

                                     parse : image -¿ ast
                                     parseAtom : lexical˙atom -¿ ast
exception Associativity
local
  exception Impossible
  fun parse’ (BOT, e, EOI) = e
    | parse’ (stack, e, ipts) =

⟨shift or reduce, depending on relative precedence of ⊕_{stack} and ⊕_{ipts}⟩

  and parse (IMAGE(a, tail)) = parse’(BOT, parseAtom a, tail)
  and parseAtom (LEX_ATOM a) = ATOM a
    | parseAtom (PARENS im) = parse im
in
  val parse = parse
end

The ML code uses two new constructs, and it shows a new way to use pattern matching. The declaration

local private-declarations in public-declarations end hides information. Private declarations are

invisible outside the local construct; only public declarations escape. The connective and is used with fun to de^ne a nest of mutually recursive functions. The ^rst match in the de^nition of parse’ handles the case when both the stack and the input are empty. The next match handles all other cases; this works because a variable in a binding matches any datatype constructor.

8 NORMAN RAMSEY

The interesting part of the parser is deciding whether to shift or reduce. Shifting pushes e and ⊕_{ipts} onto the stack. Reducing pops ⊕_{stack} and the expression below it, creating a new APP node with operator ⊕_{stack}. Shifting ⊕_{ipts} guarantees that ⊕_{ipts} will be reduced before ⊕_{stack}, and therefore that ⊕_{ipts} will eventually be a right descendant of ⊕_{stack}. Reducing guarantees that ⊕_{stack} will eventually be a left descendant of ⊕_{ipts}. The parser uses noparens to choose whichever alternative is correct without parentheses. The choice is deterministic because the conditions for shifting and reducing are mutually exclusive; the proof of correctness relies on that fact. ⟨shift or reduce, depending on relative precedence of ⊕_{stack} and ⊕_{ipts}⟩≡

if noparens(irator ipts, srator stack, RIGHT) then (* shift *)
  case ipts
    of INFIX ($, a, ipts’) =>
         parse’(PUSH(stack, e, $), parseAtom a, ipts’)
     | EOI => raise Impossible (* EOI has lowest precedence *)
else if noparens(srator stack, irator ipts, LEFT) then (* reduce *)
  case stack
    of PUSH (stack’, e’, $) => parse’(stack’, APP(e’, $, e), ipts)
     | BOT => raise Impossible (* BOT has lowest precedence *)
else
  raise Associativity
    (* saw a + b + c, for some nonassociative + *)

Pattern matching can be used in case statements as well as in function de^nitions. Here it is used to extract the ^rst two inputs or the expression and operator on top of the stack. Because srator and

irator return minrator for BOT and EOI, the parser shifts when the stack is empty and reduces

when there are no more inputs. Therefore, the lines that raise Impossible are never executed; they are included only to keep the compiler from complaining about cases that are not covered.

The parser uses two different kinds of recursion. The direct recursive calls from parse’ to itself are all tail calls; such calls are ML programmers’ way of writing loops. In an imperative language,

parse’ would be written with a loop, and it would modify three variables: the stack, the current ex-

pression, and the inputs. When the stack and inputs were empty, it would exit the loop by returning the current expression. An imperative version of parse’ would keep the indirect recursive call through

parseAtom and parse.

Unparsing in�x expressions

Before getting into the details of unparsing, we de^ne functions that make images from atoms and put parentheses around images.

⟨in�x⟩+≡ image : string -¿ image

                                             parenthesize : image -¿ image
  fun image a = IMAGE (LEX_ATOM a, EOI)
  fun parenthesize im = IMAGE (PARENS im, EOI)
infixImage combines two images by putting an in^x operator between them. It uses append as an

auxiliary function.

⟨in�x⟩+≡ append : image’ * image’ -¿ image’

                              infixImage : image * rator * image -¿ image
local
  fun append (EOI, image’) = image’
    | append (INFIX($, a, tail), image’) =

        INFIX($, a, append(tail, image’))
in
  fun infixImage (IMAGE(a, tail), $, IMAGE(a’, tail’)) =
        IMAGE(a, append(tail, INFIX($, a’, tail’)))
end

To unparse an expression, we produce a fragment containing not only the image of the expression, but also the lowest-precedence operator used in the expression. That operator tells us enough to decide whether to parenthesize that expression when it is used. ⟨in�x⟩+≡

type fragment = rator * image

The top-level operators of an image are the operators that appear outside of parentheses. If an image has top-level operators, then the top-level operators of least precedence must all have the same associativity, which must be LEFT or RIGHT. Otherwise, one of those operators would have to be in parentheses, which contradicts the assumption that they are top-level operators.

The function bracket(frag, side, rator) parenthesizes a fragment frag that is to appear next to an operator rator on the side labelled side:

⟨in�x⟩+≡ bracket : fragment * associativity * rator -¿ image

fun bracket((inner, image), side, outer) =
  if noparens(inner, outer, side) then image else parenthesize image

Given the ability to bracket fragments, unparsing is straightforward. We create another sentinel operator

maxrator, having precedence maxprec, which must be higher than the precedence of any true

operator, so we can use it in fragments made from atoms.

⟨in�x⟩+≡ unparse’ : ast -¿ rator * image

                                          unparse : ast -¿ image
local
  val maxrator = ("<max-precedence sentinel>", maxprec, NONASSOC)
  fun unparse’ (ATOM a) = (maxrator, image a)
    | unparse’ (APP(l, $, r)) =
        let val l’ = bracket (unparse’ l, LEFT, $)
            val r’ = bracket (unparse’ r, RIGHT, $)
        in ($, infixImage(l’, $, r’))
        end
in
  fun unparse e =
    let val ($, im) = unparse’ e
    in im
    end
end

Using bracket maintains the precedence and associativity invariants of the fragments. unparse ^rst computes a fragment, then discards the operator, leaving only the image.

10 NORMAN RAMSEY

Proof of correctness

Proving the simple unparser correct helps clarify how the parser and unparser work together. The result giving the most insight is Proposition 3, which gives the properties of the top-level operators produced during unparsing. Most of the proof is omitted from the body of the paper; it appears instead in Appendix B.

Proposition 1 parse (unparse e) = e.

This proposition follows from Proposition 2.

Proposition 2 (Correctness of unparsing in context) If e is any expression, and if a and tail are chosen such that unparse e = IMAGE(a, tail), and if s is a parser stack and i an input sequence of type image’ such that for all top-level operators ⊕^0 in tail,

noparens(⊕^0; srator s; RIGHT) and noparens(⊕^0; irator i; LEFT);

then parse’ (s, parseAtom a, append(tail, i)) = parse’ (s, e, i).

In other words, even when e is a fragment of a larger expression, we can parse the concrete syntax (a,

tail) after having parsed a pre^x (stack s) and before parsing a suf^x (inputs i), and in this context

we recover the original expression e, leaving stack s and inputs i to be processed. The conditions involving noparens say that the top-level operators in unparse e bind at least as tightly as ⊕_{stack}, which is on top of the stack, and at least as tightly as ⊕_{ipts}, which follows e; furthermore, when the binding powers are the same, the associativity forces the parser to shift or reduce as appropriate.

De�nition 1 (Binds as tightly as) An operator ⊕ binds as tightly as another operator rator if and only if

precedence⊕ > precedencerator ∨precedence⊕ = precedencerator ∧ associativity⊕ = associativityrator:

This relation is like noparens, except it ignores the relative positions of operators of the same precedence and associativity. This weakening of noparens yields a re~exive, transitive, binary relation, which is in fact a slight variation on Şhas precedence at least as great.Ť

The proof of Proposition 2 is by structural induction on e. The base case is straightforward. To get the idea of the proof of the induction step, we use textual juxtaposition to write expressions, inputs, and parser stacks. If e = l ⊕ r, we can write unparse e = a_lt_l ⊕ a_rt_r, where a’s are atoms and

t’s (tails) are sequences of atoms and operators. Writing parse^0(s; parseAtom a_l; t_l ⊕ a_rt_ri)

for the state of the parser, we prove Proposition 2 by showing

parse^0(s; parseAtom a_l; t_l ⊕ a_rt_ri)

= (by the induction hypothesis applied to l)

parse^0(s; l; ⊕a_rt_ri)

= (by hypothesis, noparens(⊕; srator s; RIGHT), and the parser shifts)

parse^0(sl⊕; parseAtom a_r; t_ri)

= (by the induction hypothesis applied to r) parse^0(sl⊕; r; i)

= (by hypothesis, noparens(⊕; irator i; LEFT), and the parser reduces)

parse^0(s; l ⊕ r; i) = parse^0(s; e; i)

The details involve showing that the preconditions hold and it is safe to apply the induction hypothesis. We need facts about the results returned by unparse and unparse’, as stated in the next two propositions.

Proposition 3 If an expression e = APP(l; ⊕; r), then we can choose l^0 and r^0 as in unparse^0, so that unparse e = l^0 ⊕ r^0.

1. If ⊕ is non-associative, then every top-level operator in l^0 and r^0 has a precedence greater than that of ⊕.

2. If ⊕ is left-associative, then every top-level operator in r^0 has a precedence greater than that of ⊕. 3. If ⊕ is right-associative, then every top-level operator in l^0 has a precedence greater than that of ⊕.

Proposition 4 For any expression e, unparse’ e is properly covered, as de�ned below.

De�nition 2 (Covered, properly covered) A fragment frag = (rator, im) is covered if and only if every top-level operator in im binds as tightly as rator. A covered fragment is properly covered if and only if rator = maxrator or there is at least one top-level operator in im.

The intuition behind a covered fragment (rator, im) is that low-precedence operators in im, which do not bind as tightly as rator, are safely ŞcoveredŤ with parentheses. Proper coverage avoids pathology for fragments that don’t have any top-level operators.

Proposition 3 is proved by case analysis on the result returned by bracket. Proposition 4 is proved by structural induction on the expression unparsed. The proof uses this lemma:

Lemma 1 (Bracketing lemma) If fragment f is properly covered, then for any operator rator and associativity a, bf = (rator, bracket(f, a, rator)) is covered.

The lemma is proved by case analysis on whether bracket parenthesizes its fragment. The proof uses transitivity of Şbinds as tightly as.Ť

The proof is not as simple as it would have been if parse and unparse had been designed in tandem with a proof, but parse and unparse were designed to have simple implementations, not a simple proof. One reason the proof is complex is that the functions are not full inverses; because parsing leaves no record of redundant parentheses, unparse(parse e) ≠ e. Moreover, although the functions are partial inverses, the computations are not inverses. The parser is an LR parser, working left to right with an explicit stack. The unparser works not left to right but bottom-up, with an implicit stack created by recursive calls.

An alternative approach to unparsing might begin with a parser and attempt to derive an unparser using the program-transformation techniques known as program inversion.^{12-14} If successful, this attempt would produce an unparser that would be correct by construction, but it would look substantially different from the one presented here. Program-inversion rules have been developed for simple programming calculi, not real programming languages, so the unparser would have to be translated into a programming language in order to be executed. The unparsers in this paper are executable directly, and they are designed to be easy to understand, even if that design makes the proof of correctness harder to understand.

12 NORMAN RAMSEY

4. ADDING PREFIX AND POSTFIX OPERATORS

Simple in^x expressions are inadequate for many real programming languages, including C and ML. In this section, we add unary pre^x and post^x operators, as well as in^x operators of arbitrary arity. Moreover, we use the Standard ML modules system to generalize our idea of what an image is, so we can produce output for a prettyprinter, not just text.

Fixity distinguishes pre^x and post^x operators from in^x operators. Only in^x operators have associativity. ⟨assoc.sml⟩≡

structure Assoc = struct
  datatype associativity = LEFT | RIGHT | NONASSOC
  datatype fixity = PREFIX | POSTFIX | INFIX of associativity
end

From these de^nitions, the possible values of type fixity are PREFIX, POSTFIX, INFIX LEFT,

INFIX RIGHT, and INFIX NONASSOC.

Having pre^x and post^x operators means that precedence alone no longer determines the correct parse, because the order of two pre^x or of two post^x operators overrides precedence. For example, if ++ and * are pre^x operators and p is an atom, then no matter what the precedences of ++ and *,

++*p and *++p are both correct and unambiguous, equivalent to ++(*p) and *(++p), respectively.

In^x operators of arbitrary arity broaden our view of non-associative operators. For example, in both ML and C, the following three expressions are all legal, but all different:

f((a; b); c) ≠ f(a; b; c) ≠ f(a; (b; c))

The comma operator is an n-ary, in^x, non-associative operator.

Operators now have a concrete representation, a precedence, and a ^xity. ⟨general types⟩≡

type precedence = int
type rator = Atom.atom * precedence * Assoc.fixity

In expressions, operators may be unary, binary, or n-ary. ⟨type of expressions⟩≡

datatype ast = ATOM of Atom.atom
             | UNARY of rator * ast
             | BINARY of ast * rator * ast
             | NARY of rator * ast list

List types are built in to ML. Type ’a list means Şa list of values of type ’a,Ť for any type ’a. It is convenient to treat NARY(⊕, [e]) as equivalent to e for any ⊕; [e] is the singleton list containing e.

In this new implementation, atomic components of expressions are no longer limited to strings, but may be any values of any type. That type is given the name Atom.atom, but it is not otherwise speci^ed. The user of the unparser can pick any convenient type, e.g., prettyprinter directives. The structure Atom is a parameter to the unparser module. It must provide the following types and values: ⟨properties of structure Atom⟩≡

type atom (* representation of operator or atomic expression *)
val parenthesize : atom list -> atom

The unparser uses Atom.parenthesize to parenthesize a list of atoms.

The chunk ⟨properties of structure Atom⟩ exploits a feature of the Standard ML modules system; in a signature, we can name types and values without giving their de^nitions. The unparser module is parameterized by a module called Atom, and it works with any Atom that provides a type atom, a function parenthesize, and the other values listed in ⟨properties of structure Atom⟩. This parameterization makes the unparser general and easy to reuse.

To make this general capability as useful as possible, we make it possible to have the unparser add ŞdecorationsŤ to list of atoms. We do so by adding another kind of expression. ⟨type of expressions⟩+≡

| DECORATE of (Atom.atom list -> Atom.atom) * ast

The DECORATE constructor attaches a Şdecoration functionŤ to an expression. The decoration function must transform a list of atoms into a single atom in a way that is indetectible by the parser. For example, a decoration function might add values of type Atom.atom that indicate changes in indentation, color, or font of the unparsed text. These values would appear to the parser as white space.

Finally, the user of the unparser must supply minimum and maximum precedences, which are used to de^ne the sentinels minrator and maxrator. ⟨properties of structure Atom⟩+≡

val minprec : int (* precedence below that of any operator *)
val maxprec : int (* precedence above that of any operator *)
val bogus : atom (* bogus atom used to make sentinel operators *)

In many applications, minprec and maxprec can be computed automatically.

The datatype ast and the structure Atom form the external interfaces to the unparser. In the implementation, images are more generalŮoperators and atomic terms no longer alternate, but they are still properly parenthesized:

image ⇒ { operator | | lexical-atom | | decorated-image}

lexical-atom ⇒ atom | | ( image )

The corresponding ML types are: ⟨general types⟩+≡

datatype image = IMAGE of lexeme list
and lexeme = RATOR of rator
                 | EXP of lexical_atom
                 | DECORATED of (Atom.atom list -> Atom.atom) * image
and lexical_atom = LEX_ATOM of Atom.atom
                 | PARENS of image

where lexeme corresponds to ( operator | | lexical-atom | | decorated-image).

14 NORMAN RAMSEY

Type image is internal; the unparser returns a list of atoms. It uses flatten to convert an image into such a list. It uses Atom.parenthesize for parenthesization.

⟨general⟩≡ image : image -¿ Atom.atom list

                                 lexeme : lexeme -¿ Atom.atom
                                 lexical˙atom : lexical˙atom -¿ Atom.atom
                                 flatten : image -¿ Atom.atom list
local
  fun image (IMAGE l) = map lexeme l
  and lexeme (RATOR (atom, _, _)) = atom
    | lexeme (EXP a) = lexical_atom a
    | lexeme (DECORATED (f, im)) = f (image im)
  and lexical_atom (LEX_ATOM atom) = atom
    | lexical_atom (PARENS im) = Atom.parenthesize (image im)
in
  val flatten = image
end

Each local function computes an atom or list of atoms that corresponds to a value of one type. By programming convention, the name of each function is the name of the type it operates on. Because

local hides these functions, flatten is the only name visible outside this code chunk.

Parenthesizing general expressions

As in the simpler case, development of an unparser begins with the invariants that apply to an abstract syntax tree produced by parsing an image without parentheses. There are now three cases to consider: an operator may be the left child of a binary in^x operator, the right child of a binary in^x operator, or the only child of a unary pre^x or post^x operator. ⊕

⊕_l

fi fi fi fi (a)

⊕

⊕_r (b)

L L L L

⊕

⊕^0 (c) As before, the rules for parenthesization are encapsulated in a noparens function. Because these trees are obtained by parsing parenthesis-free syntax, in case (a), noparens(⊕_l; ⊕; LEFT); in case (b),

noparens(⊕_r; ⊕; RIGHT); and in case (c), noparens(⊕^0; ⊕; NONASSOC).

⟨general⟩+≡ noparens : rator * rator * Assoc.associativity -¿ bool

fun noparens(inner as (_, pi, fi), outer as (_, po, fo), side) =
  pi > po orelse (* (a), (b), or (c) *)
  case (fi, side)
    of (POSTFIX, LEFT) => true (* (a) *)
     | (PREFIX, RIGHT) => true (* (b) *)
     | (INFIX LEFT, LEFT) =>
          pi = po andalso fo = INFIX LEFT (* (a) *)
     | (INFIX RIGHT, RIGHT) =>
          pi = po andalso fo = INFIX RIGHT (* (b) *)
     | (_, NONASSOC) => fi = fo (* (c) *)
     | _ => false

These rules can be understood as follows:

• If the inner operator has higher precedence, parentheses are not needed. • If a post^x operator appears to the left of an in^x operator (a), it always applies to the preceding expression, and parentheses are not needed. Similarly when a pre^x operator appears to the right of an in^x operator (b).

• In case (a), if ⊕_l has lower precedence than ⊕, then parentheses are needed, but if the two have the same precedence, then parentheses can be avoided, provided both are left-associative. A similar argument applies to case (b) when both operators are right-associative.

• Finally, in case (c), if the precedence of ⊕^0 is not greater than the precedence of ⊕, then parentheses can be avoided if and only if both operators have the same ^xity, i.e., both are pre^x operators or both are post^x operators. This is the case in the example of ++*p used to introduce this section.

These rules for parenthesization, as embodied in the noparens predicate, are the key to understanding both parsing and unparsing algorithms. The noparens function is used in an unparser, which appears below, and in a parser, which appears in Appendix A.

As before, the unparser needs auxiliary functions to manipulate images. Because an image is now simply a list of lexemes, these functions are a bit simpler.

⟨general⟩+≡ image : Atom.atom -¿ image

                            parenthesize : image -¿ image
                            infixImage : image * rator * image -¿ image
                            prefixImage : rator * image -¿ image
                            postfixImage : image * rator -¿ image
fun image a = IMAGE [EXP (LEX_ATOM a)]
fun parenthesize image = IMAGE [EXP (PARENS image)]
fun infixImage (IMAGE l, $, IMAGE r) = IMAGE (l @ RATOR $ :: r)
fun prefixImage ( $, IMAGE r) = IMAGE ( RATOR $ :: r)
fun postfixImage(IMAGE l, $ ) = IMAGE (l @ RATOR $ :: [])

where @, ::, and [] are list append, cons, and the empty list, which are all prede^ned in ML.

We use the same bracket to parenthesize sub-expressions.

⟨general⟩+≡ bracket : fragment * Assoc.associativity * rator -¿ image

type fragment = rator * image
fun bracket((inner, image), side, outer) =
  if noparens(inner, outer, side) then image else parenthesize image

The structure of the unparser is as before:

⟨general⟩+≡ unparse : ast -¿ Atom.atom list

local
  val maxrator = (Atom.bogus, Atom.maxprec, INFIX NONASSOC)
  exception Impossible

⟨function unparse’, to map an expression to a fragment⟩

in
  fun unparse e =
    let val ($, im) = unparse’ e
    in flatten im
    end
end

16 NORMAN RAMSEY

The unparsing function itself has new cases: the unary operator, the n-ary operator, and the decorated expression. The unary operator appears before or after its operand, depending on its ^xity.

⟨function unparse’, to map an expression to a fragment⟩≡ unparse’ : ast -¿ fragment

fun unparse’ (ATOM a) = (maxrator, image a)
  | unparse’ (BINARY(l, $, r)) =
      let val l’ = bracket (unparse’ l, LEFT, $)
          val r’ = bracket (unparse’ r, RIGHT, $)
      in ($, infixImage(l’, $, r’))
      end
  | unparse’ (UNARY($, e)) =
      let val e’ = bracket (unparse’ e, NONASSOC, $)
          val (_, _, fixity) = $
      in ($, case fixity of PREFIX => prefixImage ($, e’)
                           | POSTFIX => postfixImage(e’, $)
                           | INFIX _ => raise Impossible)
      end
  | unparse’ (NARY(_, [])) = raise Impossible
  | unparse’ (NARY($, [e])) = unparse’ e
  | unparse’ (NARY($, e::es)) =
      let val leftmost = bracket(unparse’ e, LEFT, $)
          fun addOne(r, leftPrefix) =
                infixImage(leftPrefix, $,
                           bracket(unparse’ r, RIGHT, $))
      in ($, foldl addOne leftmost es)
      end
  | unparse’ (DECORATE (f, e)) =
      let val ($, im) = unparse’ e
      in ($, IMAGE [DECORATED (f, im)])
      end

The interesting case for the n-ary operator is covered by the pattern NARY($, e::es), which represents the operator $ applied to more than one operand. The code inserts $ between the operands. The ^rst operand, e, is used to make leftmost, an initial Şleft pre^x.Ť The addOne function extends the left pre^x, adding operator and operand on the right. The built-in foldl function walks the list es, using addOne to extend the left pre^x with the remaining operands. Because both LEFT and

RIGHT are used in the arguments to bracket, operator must be non-associative for the operands

to be parenthesized properly. Of course, no other associativity makes sense, because a left- or rightassociative operator can always be treated as an in^x binary operator. The n-ary case is needed only for non-associative operators.

5. USING THE UNPARSER

Before the unparser can be used, it has to be instantiated with a suitable Atom module. Once the user creates an Atom module, he or she combines it with the unparser’s implementation to create

Unparser, an instance of the unparser that is specialized to that particular Atom. This instance pro-

vides the datatype ast, with its contructors ATOM, UNARY, etc. It also provides an unparsing function. ⟨values exported by the unparser⟩≡

val unparse : ast -> Atom.atom list

Once the unparser is instantiated, it can be used in a real tool, where unparsing is usually one of a series of transformations. In a typical application generator, a program being generated passes through several representations:

• an application-speci^c internal representation, • an unparsing tree of type Unparser.ast, • a list of atoms of type Atom.atom list, which typically includes prettyprinting directives as well as text, and • the ^nal program, represented as nicely indented text in the target language.

This section discusses transformation from an internal representation into an unparsing tree, using examples from the New Jersey Machine-Code Toolkit.^{15} The ^rst part of this paper discusses transformation from an unparsing tree to a list of atoms. References 1 and 2 discuss transformation of prettyprinting directives to indented text.

A preliminary step in creating unparsing trees is assigning precedence and associativity to the operators of the target language. This can be done easily and reliably by putting operators in a data structure that indicates precedence implicitly and ^xity (which includes associativity) explicitly. A fragment of the structure used for C is

   .
   .
(L, [">>", "<<"]),
(L, ["+", "-"]),
(L, ["%", "/", "*"]),
   .
   .

where L is an abbreviation for Assoc.INFIX Assoc.LEFT, Şin^x leftŤ being one of the ^ve possible ^xities. Precedence increases as we move down the structure. This data structure is transformed into functions Cprec and Cfixity that return the precedence and ^xity of given operators. It is also used to compute the values of minprec and maxprec, which are needed to create an unparser.

Supplying minprec and maxprec to the unparser may seem awkward. One could supply instead a list of all operators to be used, letting the unparser compute minprec and maxprec. This technique seems unnecessarily restrictive; it would limit the unparser not only to a single kind of atom, but also to a single set of operators. As written, one unparser can be (and is) used to emit code in different target languages, using different operator sets.

Another way to eliminate minprec and maxprec would be to use a different representation of precedence inside the unparser, e.g., to represent precedence by an ML datatype that included values +∞ and −∞ in addition to all the integers. In a production environment, eliminating minprec and

maxprec from the interface would justify a more complex implementation. For expository purposes,

however, another user-de^ned datatype and special comparison functions would distract readers from the explanation of unparsing.

The ^xities of operators can be surprising. For example, the C operators . and ->, which are used to select members from structures and unions, are super^cially binary in^x operators, but their right-hand ŞargumentsŤ are not expressions but identi^ers, and semantically they are unary operators. The Toolkit in fact treats them as unary post^x operators, choosing their representations dynamically according to what member is being selected. For example, to create an unparsing tree corresponding to the selection of member tag from structure e, it uses the function ⟨tree-creation functions⟩≡ dot : Unparser.ast * string -¿ Unparser.ast

fun dot (e, tag) =
  Unparser.UNARY((pptext ("." Ĺ tag), Cprec ".", Assoc.POSTFIX), e)

18 NORMAN RAMSEY

Here the ML Ĺ operator is string concatenation. The pptext function takes a string and converts it to an Atom.atom, making it intelligible to the prettyprinter.

Handling mix^x operators may involve unparsing some trees during the construction of other trees. For example, to translate an array-indexing expression into C, one must put the subscript in square brackets. The subscript is unparsed independently and turned into an atom by applying pplist. It is then wrapped in square brackets, and the indexing operation itself is represented by the empty string, but with proper precedence and associativity, which, for example, enable the unparser to produce either

*p[n] or (*p)[n] as required. A similar trick is used to unparse function calls after applying an

n-ary comma operator to the arguments. ⟨tree-creation functions⟩+≡

                    subscript : Unparser.ast * Unparser.ast -¿ Unparser.ast
  fun subscript(array, index) =
        let val index = pplist (Unparser.unparse index)
            val index = pplist [pptext "[", index, pptext "]"]
        in Unparser.BINARY ( array
                            , (pptext "", Cprec "subscript", L)
                            , Unparser.ATOM index
                            )
        end
pplist concatenates atoms; it is also used to de^ne Atom.parenthesize:

fun parenthesize l = pplist [pptext "(", pplist l, pptext ")"]

6. USE AND AVAILABILITY

The unparser described in this paper is used in two application generators.^{6, 15} Because the paper is a literate program, the code used in the generators is extracted directly from the source text of the paper, and it is available from the author’s Web site, http://www.cs.virginia.edu/Ÿnr. Practitioners who can’t use this code directly can build unparsers by implementing the noparens predicate and writing unparse as a bottom-up tree walk. Appendix A of this paper presents the inverse of the general unparser, i.e., a parser that handles pre^x and post^x as well as in^x operators, all of dynamically chosen precedence and associativity. This parser, which is used in one of the application generators, is also available online. Appendix B presents the full proof of correctness of the simple unparser.

ACKNOWLEDGEMENTS

This work was supported in part by NSF grant ASC-9612756 and by the US Department of Defense under contract MDA904-97-C-0247 and DARPA order number E381. The views expressed are my own and should not be taken to be those of the DoD.

I thank Referee A for demanding improvements in presentation.

APPENDIX: A. A PARSER FOR GENERAL EXPRESSIONS

The parser shown here is the inverse of the unparser in the body of the paper. It handles pre^x, post^x, and in^x operators. It also supports concrete syntax in which concatenation of two expressions signi- ^es the application of a ŞjuxtapositionŤ operator to the two expressions. Like other binary operators, juxtaposition has precedence and associativity. For example, in Standard ML, juxtaposition denotes function application, it associates to the left, and it has a precedence higher than that of any explicit in^x operator.^4 In awk, juxtaposition denotes string concatenation, and it has a precedence higher than the comparison operators but lower than the arithmetic operators.^{16}

The parser handles juxtaposition by inserting an implicit Şjuxtaposition operatorŤ whenever two expressions are juxtaposed in the input. (The unparser handles juxtaposition by using a binary in^x operator whose representation is white space.) The user supplies a speci^cation of this operator as part of the structure Atom. The speci^cation includes an associativity, not a ^xity, because juxtaposition is in^x by de^nition.

⟨properties of structure Atom⟩+≡

val juxtapositionSpec : atom * int * Assoc.associativity

Internally, the parser converts this speci^cation into an in^x operator. ⟨general parser⟩≡

local
  val (rep, prec, assoc) = Atom.juxtapositionSpec
in
  val juxtarator = (rep, prec, Assoc.INFIX assoc)
end

The parser’s main data structure is a stack of operators and expressions satisfying these invariants:

1. No post^x operator appears on the stack.

2. If a binary operator is on the stack, its left argument is below it.

3. An expression on the stack is preceded by a list of pre^x operators, then either an in^x operator or the bottom of the stack.

These invariants are only partially enforced by the type system. All operators have the same type, so

prefixop and infixop are only hints.

⟨general parser⟩+≡

type infixop = rator
type prefixop = rator
datatype stack = BOT
               | BIN of stack * prefixop list * ast * infixop
type stack_top = (prefixop list * ast)

The sentinel operator minrator appears on the bottom of the stack:

⟨general parser⟩+≡ srator : stack -¿ rator

val minrator = (Atom.bogus, Atom.minprec, INFIX NONASSOC)
fun srator (BIN (_, _, _, $)) = $
  | srator BOT = minrator

20 NORMAN RAMSEY

The parser’s input looks something like this:

image ⇒ { pre�x } lexical-atom { post�x } {[ in�x ] { pre�x } lexical-atom { post�x }} In parsing, we assume that we can distinguish pre^x, post^x, and in^x operators. If juxtaposition is forbidden, these restrictions can be relaxed somewhat; it is then suf^cient to be able to distinguish post^x from in^x operators. The adjustments needed are left as an exercise. isPrefix identi^es pre^x operators.

⟨general parser⟩+≡ isPrefix : rator -¿ bool

fun isPrefix (_, _, f) = f = PREFIX

The parser uses two functions, parsePrefix and parsePostfix, depending on whether it is scanning before or after an atomic input. The overall structure of the parser is

⟨general parser⟩+≡ parse : image -¿ ast

exception ParseError of string * rator list
local
  exception Impossible

⟨general parsing functions⟩

in
  val parse = parse
end

There are many cases. parsePrefix maintains a list of pre^x operators, which is saves until it reaches an atomic input. It is not safe to reduce the pre^x operators, because the atomic input might be followed by a post^x operator of higher precedence. If parsePrefix encounters a non-pre^x operator or end of ^le, the input is not correct. When parsePrefix encounters an atomic input, it passes the input and the saved pre^xes to parsePostfix. ⟨general parsing functions⟩≡

                 parsePrefix : stack * prefixop list * lexeme list -¿ ast
fun parsePrefix(stack, prefixes, RATOR $ :: ipts’) =
      if isPrefix $ then
        parsePrefix(stack, $ :: prefixes, ipts’)
      else
        raise ParseError("%s is not a prefix operator", [$])
  | parsePrefix(_, _, []) =
      raise ParseError ("premature end of expression", [])
  | parsePrefix(stack, prefixes, EXP a :: ipts’) =
      parsePostfix(stack, (parseAtom a, prefixes), ipts’)
  | parsePrefix(stack, prefixes, DECORATED (_, IMAGE l) :: ipts’) =
      parsePrefix(stack, prefixes, l @ ipts’)

If it encounters decorated lexemes, parsePrefix ignores the decoration.

In addition to the stack and the input, parsePostfix maintains a current expression e, and a list prefixes containing unreduced pre^x operators that immediately precede e. The idea is that operators arriving in the input must be tested either against the nearest unreduced pre^x operator, or if there are no unreduced pre^x operators, against the operator on the top of the stack. In the ^rst case, the nearest unreduced pre^x operator is called $, and it may be compared with an operator irator from the input. If $ has higher precedence, it is reduced, by using UNARY to apply it to the current expression e, and it becomes the new expression. Otherwise, the parser’s behavior depends on the ^xity of irator; it may reduce irator, shift it, or insert juxtarator. ⟨general parsing functions⟩+≡

        parsePostfix : stack * (ast * prefixop list) * lexeme list -¿ ast
and parsePostfix(stack, (e, $ :: prefixes),
                 ipts as RATOR (irator as (_, _, ifixity)) :: ipts’) =
      if noparens($, irator, NONASSOC) then (* reduce prefix $ *)
        parsePostfix(stack, (UNARY($, e), prefixes), ipts)
      else if noparens(irator, $, NONASSOC) then
        (* irator has higher precedence *)
        case ifixity
          of POSTFIX => (* reduce postfix rator *)
               parsePostfix(stack,
                            (UNARY(irator, e), $ :: prefixes),
                            ipts’)
           | INFIX _ => (* shift, look for prefixes *)
               parsePrefix(BIN(stack, $ :: prefixes, e, irator),
                           [], ipts’)
           | PREFIX => ⟨insert juxtarator⟩
      else
        raise ParseError
            ("can’t parse (%s e %s); " Ĺ
             "operators have equal precedence", [$, irator])

If parsePostfix encounters an atom in the input, it inserts juxtarator no matter what the state of unreduced pre^xes, since consuming atoms is the purview of parsePrefix. ⟨general parsing functions⟩+≡

| parsePostfix(stack, (e, prefixes), ipts as EXP _ :: _) =

⟨insert juxtarator⟩ The insertion itself is straightforward. ⟨insert juxtarator⟩≡

parsePostfix(stack, (e, prefixes), RATOR juxtarator :: ipts)

22 NORMAN RAMSEY

The other major case for parsePostfix occurs when there are no more unreduced pre^x operators. In that case, comparison must be made with srator stack, the operator on top of the stack. ⟨general parsing functions⟩+≡

| parsePostfix(stack, (e, prefixes as []),
               ipts as RATOR (irator as (_, _, ifixity)) :: ipts’
              ) =
    if noparens(srator stack, irator, LEFT) then
      (* reduce infix on stack *)
      case stack
        of BIN (stack’, prefixes, e’, $) =>
             parsePostfix(stack’, (BINARY(e’, $, e), prefixes),
                          ipts)
         | BOT => raise Impossible (* BOT has lowest prec *)
    else if noparens(irator, srator stack, RIGHT) then
      case ifixity
        of POSTFIX => (* reduce postfix rator *)
             parsePostfix(stack, (UNARY(irator, e), []), ipts’)
         | INFIX _ => (* shift, look for prefixes *)
             parsePrefix(BIN(stack, [], e, irator), [], ipts’)
         | PREFIX => ⟨insert juxtarator⟩
    else
       raise ParseError ("%s is non-associative", [irator])

As above, decorations are ignored. ⟨general parsing functions⟩+≡

| parsePostfix (stack, (e, prefixes),
                DECORATED (_, IMAGE l) :: ipts’) =
    parsePostfix(stack, (e, prefixes), l @ ipts’)

When the input is exhausted, the parser reduces ^rst the pre^xes, then the stack, until it reaches the result.

⟨general parsing functions⟩+≡

| parsePostfix(stack, (e, $ :: prefixes), []) =
    (* reduce prefix $ *)
    parsePostfix(stack, (UNARY($, e), prefixes), [])
| parsePostfix(BIN (stack’, prefixes, e’, $), (e, []), []) =
    (* reduce stack *)
    parsePostfix(stack’, (BINARY(e’, $, e), prefixes), [])
| parsePostfix(BOT, (e, []), []) = e

We complete the parser with functions that parse an entire input, or an input in parentheses.

⟨general parsing functions⟩+≡ parse : image -¿ ast

                                          parseAtom : lexical˙atom -¿ ast
and parse(IMAGE(l)) = parsePrefix(BOT, [], l)
and parseAtom (LEX_ATOM a) = ATOM a
  | parseAtom (PARENS im) = parse im

APPENDIX: B. PROOFS OF CORRECTNESS Proofs are written in a goal-directed, calculational style. A goal-directed proof begins with a conclusion to be proved, then reasons backwards towards premises. The backwards arrow ⇐ is pronounced Şis implied by.Ť Proposition 1 parse (unparse e) = e.

Proof Rewrite parse (unparse e) into an expression more easily proved equal to e.

parse (unparse e)

= (by choosing suitable a and tail for the result of unparse e)

parse (IMAGE (a, tail))

= (by the de^nition of parse)

parse’ (BOT, parseAtom a, tail)

= (by the de^nition of append)

parse’ (BOT, parseAtom a, append(tail, EOI))

= (by Proposition 2)

parse’(BOT, e, EOI)

= (by de^nition of parse’)

e

3 Proposition 2 (Correctness of unparsing in context) If e is any expression, and if a and tail are chosen such that unparse e = IMAGE(a, tail), and if s is a parser stack and i an input sequence of type image’ such that for all top-level operators ⊕^0 in tail,

noparens(⊕^0; srator s; RIGHT) and noparens(⊕^0; irator i; LEFT);

then parse’ (s, parseAtom a, append(tail, i)) = parse’ (s, e, i). Proof By structural induction on e. If e is an atom, write e = ATOM x, and

parse’ (s, parseAtom a, append(tail, i))

= (because by assumption unparse e = IMAGE (LEX ATOM x, EOI))

parse’ (s, parseAtom (LEX ATOM x), append(EOI, i))

= (by de^nitions of parseAtom and append)

parse’ (s, ATOM x, i)

= (by assumption)

parse’ (s, e, i)

24 NORMAN RAMSEY For the induction step, e must have the form e = APP(l; ⊕; r), and we can choose l^0 and r^0 as in unparse^0, so unparse e = infixImage(l^0; ⊕; r^0). Because both l^0 and r^0 have type image, we can write l^0 = IMAGE(a_l; t_l) and r^0 = IMAGE(a_r; t_r). By using juxtaposition to abbreviate appending, we write unparse e = a_lt_l ⊕ a_rt_r.^1 We therefore consider

parse^0(s; parseAtom a_l; t_l ⊕ a_rt_ri)

= (by applying the induction hypothesis to l^0, as justi^ed below)

parse^0(s; l; ⊕a_rt_ri)

= (by de^nition of parse’, the parser shifts, because by hypothesis, ⊕ is a top-level operator from

unparse e and therefore noparens(⊕; srator s; RIGHT))
  parse^0(sl⊕; parseAtom a_r; t_ri)

= (by applying the induction hypothesis to r^0, as justi^ed below)

parse^0(sl⊕; r; i)

= (by de^nition of parse’, the parser reduces, because by hypothesis, ⊕ is a top-level operator from

unparse e and therefore noparens(⊕; irator i; LEFT))
  parse^0(s; APP(l; ⊕; r); i)

= (by assumption)

parse^0(s; e; i)

It remains to justify applying the induction hypothesis to l^0 and r^0. By de^nition of unparse and

bracket, either
           l^0 = unparse l or l^0 = parenthesize(unparse l):

In the latter case, l^0 = IMAGE(PARENS(unparse l); EOI), so by the de^nitions of a_l and t_l, we have a_l = PARENS(unparse l) and t_l = EOI, and

parse^0(s; parseAtom a_l; t_l ⊕ a_rt_ri)

= (by assumption and by de^nition of parseAtom)

parse^0(s; parse(unparse l); ⊕a_rt_ri)

= (by the induction hypothesis applied to l, with empty stack and inputs)

parse^0(s; l; ⊕a_rt_ri)

which is the step made above.

The case where l^0 = unparse l is the interesting case.

parse^0(s; parseAtom a_l; t_l ⊕ a_rt_ri)

= (For all top-level operators ⊕^0 in t_l, noparens(⊕^0; ⊕; LEFT), and we apply the induction hypothesis with the inputs ⊕r^0 i, concluding that)

� This shorthand abbreviates unparse e = IMAGE(a_l; append(t_l; INFIX(�; a_r; t_r))).

parse^0(s; parseAtom a_l; t_l ⊕ a_rt_ri)

We must show noparens(⊕^0; ⊕; LEFT) for any top-level ⊕^0 from t_l. If ⊕ is non- or right-associative, then by Proposition 3, precedence⊕^0 > precedence⊕, and noparens(⊕^0; ⊕; LEFT) follows. If ⊕ is left-associative, then

noparens(⊕^0; ⊕; LEFT)

⇐ (by de^nition of noparens and left-associativity of ⊕)

⊕^0 binds as tightly as ⊕ ⇐ (transitivity)

⊕^0 binds as tightly as ⊕_l ∧ ⊕_l binds as tightly as ⊕ Where unparse^0 l = (⊕_l; a_lt_l) and the two clauses follow by applying Proposition 4 to l and e. 3 Proposition 3 If an expression e = APP(l; ⊕; r), then we can choose l^0 and r^0 as in unparse^0, so that unparse e = l^0 ⊕ r^0.

1. If ⊕ is non-associative, then every top-level operator in l^0 and r^0 has a precedence greater than that of ⊕.

2. If ⊕ is left-associative, then every top-level operator in r^0 has a precedence greater than that of ⊕. 3. If ⊕ is right-associative, then every top-level operator in l^0 has a precedence greater than that of ⊕.

Proof First, do case analysis on l^0 = bracket(unparse^0 l; LEFT; ⊕). If l^0 is parenthesized, then (3) and half of (1) hold vacuously because l^0 has no top-level operators. Otherwise, we can assume that unparse^0 l = (⊕_l; l^0), and

For all top-level operators ⊕^0 in l^0, precedence⊕^0 > precedence⊕ ⇐ (introducing ⊕_l)

For all top-level operators ⊕^0 in l^0, precedence⊕^0 ≥ precedence⊕_l ∧ precedence⊕_l > precedence⊕ ⇐ (de^nitions of ŞcoveredŤ and Şbinds as tightly asŤ)

Fragment (⊕_l; l^0) is covered ∧ precedence⊕_l > precedence⊕ ⇐ (de^nition of noparens)

Fragment (⊕_l; l^0) is covered ∧ noparens((⊕_l; l^0); ⊕; LEFT) ∧ associativity⊕ ≠ LEFT ⇐ (de^nition of bracket)

Fragment (⊕_l; l^0) is covered ∧ bracket((⊕_l; l^0); ⊕; LEFT) is not parenthesized ∧ associativity⊕ ≠ LEFT = (by assumption)

unparse l is covered

∧ l^0 is not parenthesized ∧ associativity⊕ ≠ LEFT

26 NORMAN RAMSEY ⇐ (Proposition 4, assumption)

associativity⊕ ≠ LEFT Which proves (3) and half of (1). A similar argument proves (2) and the other half of (1). 3 Proposition 4 For any expression e, unparse’ e is properly covered. Proof By structural induction. The base case is trivial.

For the induction step, we must prove

unparse’ e is properly covered

= (letting e = l ⊕ r and unparse e = l^0 ⊕ r^0 as above)

(⊕; l^0 ⊕ r^0) is properly covered ⇐ (⊕ is a top-level operator of l^0 ⊕ r^0)

(⊕; l^0 ⊕ r^0) is covered = (de^nition of ŞcoveredŤ)

For all top-level operators ⊕^0 in l^0 ⊕ r^0; ⊕^0 binds as tightly as ⊕ = (distributing)

For all top-level operators ⊕^0 in l^0; ⊕^0 binds as tightly as ⊕ ∧ ⊕ binds as tightly as ⊕ ∧ For all top-level operators ⊕^0 in r^0; ⊕^0 binds as tightly as ⊕ = (re~exivity of Şbinds as tightly as,Ť de^nition of ŞcoveredŤ)

(⊕; l^0) is covered ∧ (⊕; r^0) is covered = (body of unparse’)

(⊕; bracket(unparse^0 l; LEFT; ⊕)) is covered ∧ (⊕; bracket(unparse^0 r; RIGHT; ⊕)) is covered ⇐ (bracketing lemma)

unparse^0 l properly covered and unparse^0 r properly covered

which is the induction hypothesis. 3 Lemma 1 (Bracketing lemma) If fragment f is properly covered, then for any operator rator and associativity a, bf = (rator, bracket(f, a, rator)) is covered. Proof By case analysis on the result of bracket. If the result is parenthesized, then bf is trivially covered, because it contains no top-level operators.

Otherwise, let us write f = ($, im). Our goal is to show that

bf = (rator, bracket(($, im), a, rator) is covered.

⇐ (by the assumption that the result of bracket is im)

(rator, im) is covered

≡ (by De^nition 2)

For all top-level operators ⊕ in im; ⊕binds as tightly as rator

⇐ (by transitivity of Şbinds as tightly asŤ)

For all top-level operators ⊕ in im; ⊕ binds as tightly as $ ∧ $ binds as tightly as rator

⇐ (pulling second clause out of ∀ and strengthening it)

(For all top-level operators ⊕ in im; ⊕ binds as tightly as $) ∧ $ binds as tightly as rator ∧ associativity$ = a

≡ (by de^nitions of ŞcoveredŤ and noparens)

($, im) is covered

∧ noparens($, rator, a)

⇐ (by de^nition of Şproperly coveredŤ and of bracket)

($, im) is properly covered

∧ bracket(($, im), associativity$, rator) is not parenthesized

both of which are true by assumption. 3

REFERENCES

1. Derek C. Oppen, ‘Prettyprinting’, ACM Transacations on Programming Languages and Systems, 2(4), 465Ű 483 (1980).

2. John Hughes, ‘The Design of a Pretty-printing Library’, J. Jeuring and E. Meijer (eds.), Advanced Functional Programming. Springer Verlag, LNCS 925, 1995, pp. 53Ű96.

3. Donald E. Knuth, ‘Literate programming’, The Computer Journal, 27(2), 97Ű111 (1984). 4. Robin Milner, Mads Tofte, and Robert W. Harper, The De�nition of Standard ML, MIT Press, Cambridge, Massachusetts, 1990.

5. Norman Ramsey and Mary F. Fernťandez, ‘Specifying representations of machine instructions’, ACM Transactions on Programming Languages and Systems, 19(3), 492Ű524 (1997).

6. Norman Ramsey and Jack W. Davidson, ‘Machine descriptions to build tools for embedded systems’, ACM SIGPLAN Workshop on Languages, Compilers, and Tools for Embedded Systems (LCTES’98), volume 1474 of LNCS. Springer Verlag, June 1998, pp. 172Ű188.

7. Jeffrey D. Ullman, Elements of ML Programming, Prentice Hall, Englewood Cliffs, NJ, 1994. 8. Laurence C. Paulson, ML for the working programmer, Cambridge University Press, New York, second edition, 1996.

9. Matthias Felleisen and Daniel P. Friedman, The Little MLer, MIT Press, December 1997. 10. Niklaus Wirth, ‘What can we do about the unnecessary diversity of notation for syntactic de^nitions?’, Communications of the ACM, 20(11), 822Ű823 (1977).

11. Alfred V. Aho, Ravi Sethi, and Jeffrey D. Ullman, Compilers: Principles, Techniques, and Tools, Addison- Wesley, 1986.

12. David Gries, The Science of Programming, Springer-Verlag, 1981.

13. Wei Chen and Jan Tijmen Udding, ‘Program inversion: More than fun!’, Science of Computer Programming, 15(1), 1Ű13 (1990).

28 NORMAN RAMSEY

14. J. von Wright, ‘Program inversion in the re^nement calculus’, Information Processing Letters, 37(2), 95Ű100 (1991).

15. Norman Ramsey and Mary F. Fernťandez, ‘The New Jersey Machine-Code Toolkit’, Proceedings of the 1995 USENIX Technical Conference, New Orleans, LA, January 1995, pp. 289Ű302.

16. Alfred V. Aho, Brian W. Kernighan, and Peter J. Weinberger, The AWK Programming Language, Addison- Wesley, Reading, MA, 1988.